Needles in a Haystack:

Special Varieties via Small Fields

Frank-Olaf Schreyer and Fabio Tonoli

In this article we illustrate how picking points over a finite field at random can help to investigate algebraic geometry questions. In the first part we develop a program that produces random curves of genus g ≤ 14. In the second part we use the program to test Green’s Conjecture on syzygies of canonical curves and compare it with the corresponding statement for Coble self-dual sets of points. In the third section we apply our techniques to produce Calabi-Yau 3-folds of degree 17 in P^6.

Introduction

The advances in speed of modern computers and computer algebra systems gave life to the idea of solving equations by guessing a solution. Suppose M ⊂ G is a subvariety of a rational variety of codimension c. Then we expect that the probability for a point p ∈ G (F_q ) to lie in M (F_q ) is about 1/q^c. Here F_q denotes the field with q elements.

We will discuss this idea in the following setting: M will be a parameter space for objects in algebraic geometry, e.g., a Hilbert scheme, a moduli space, or a space dominating such spaces.

The most basic question we might have in this case is whether M is nonempty and whether an open part of M corresponds to smooth objects.

Typically in these cases we will not have explicit equations for M ⊂ G but only an implicit algebraic description of M , and our approach will be successful if the time required to check p /∈ M (F_q ) is sufficiently small compared to q^c. The first author applied this method first in [32] to construct some rational surfaces in P^4; see [15,11] for motivation.

In this first section we describe a program that picks curve of genus g ≤ 14 at random. The moduli spaces M_g are known to be unirational for g ≤ 13; see [33,8].

Our approach based on this result can viewed as a computer aided proof of the unirationality. Many people might object that this not a proof because we cannot control every single step in the computation. We however think that such a proof is much more reliable than a proof based on man-made computations. A mistake in a computer aided approach most often leads to an output far away from our expectation, hence it is easy to spot. A substantial improvement of present computers and computer algebra systems would give us an explicit unirational parametrization of M_g for g ≤ 13.

In the second part we apply our “random curves” to probe the consequences of Green’s conjecture on syzygies of canonical curves, and compare these results with the corresponding statements for “Coble self-dual” sets of 2g − 2 points in P^{g−2}.

In the last section we exploit our method to prove the existence of three components of the Hilbert scheme of Calabi-Yau 3-folds of degree 17 in P^6 over the complex numbers. This is one of the main results of the second author’s thesis [34, Chapter 4]. Calabi-Yau threefolds of lower degree in P^6 are easy to construct, using the Pfaffian construction and a study of their Hartshorne-Rao modules. For degree 17 the Hartshorne-Rao module has to satisfy a subtle condition. Explicit examples of such Calabi-Yau 3-folds are first constructed over a finite field by our probabilistic method. Then a delicate semi-continuity argument gives us the existence of such Calabi-Yau 3-folds over some number field.

Acknowledgments. We thank Hans-Christian v. Bothmer and Dan Grayson for valuable discussions and remarks.

Notation. For a finitely generated graded module M over the polynomial ring S = k[x_0, . . . , x_r] we summarize the numerical information of a finite free resolution 0 ← M ← F_0 ← F_1 ← . . . ← F_n ← 0

with F_i = ⊕_j S(−j)^{βij} in a table of Betti numbers, whose ij^{th} entry is

β_{i,i+j} = dim Tor^S_i (M, k)_{i+j} .

As in the Macaulay 2 command betti we suppress zeroes. For example the syzygies of the rational normal curve in P^3 have the following Betti table.

1 - - - 3 2

Note that the degrees of the entries of the matrices in the free resolution can be read off from the relative position of two numbers in consecutive columns. A pair of numbers in a line corresponds to linear entries. Quadratic entries correspond to two numbers of a square. Thus

1 - - - - 5 5 - - - - 1

corresponds to a 4 term complex with a quadratic, a linear and another quadratic map. The Grassmannian G (2, 5) in its Plücker embedding has such a free resolution.

1 How to Make Random Curves up to Genus 14

The moduli space of curves M_g is known to be of general type for g ≥ 24 and has non-negative Kodaira dimension for g = 23 by work of Harris, Mumford and Eisenbud [21,13]. For genus g ≤ 13 unirationality is known [8,33]. In this section we present a Macaulay 2 program that over a finite field F_q picks a point in M_g(F_q ) for g ≤ 14 at random.

By Brill-Noether theory [2] every curve of genus g has a linear system g^r_d of dimension r and degree d, provided that the Brill-Noether number ρ satisfies ρ := ρ(g, d, r) := g − (r + 1)(g − d + r) ≥ 0. We utilize this to find appropriate (birational) models for general curves of genus g.

1.1 Plane Models, g ≤ 10

This case was known to Severi; see [1]. Choose d = g + 2 − ⌊g/3⌋. Then ρ(g, d, 2) ≥ 0 i.e., a general curve of genus g has a plane model C^′ of degree d. We expect that C^′ has δ = ( d − 1 2 ) − g double points. If the double points are in general position, then

s = h^0(P^2, O(d)) − 3δ − 1

is the expected dimension of the linear system of curves of degree d with δ assigned double points. We have the following table:

g 1 2 3 4 5 6 7 8 9 10 11 12 ρ 1 2 0 1 2 0 1 2 0 1 2 0 d 3 4 4 5 6 6 7 8 8 9 10 10 δ 0 1 0 2 5 4 8 13 12 18 25 24 s 9 11 14 14 12 15 11 5 8 0 -10 -7

Thus for g ≤ 10 we assume that these double points lie in general position. For g > 10 the double points cannot lie in general position because s < 0. Since it is difficult to describe the special locus H_δ(g) ⊂ Hilb_δ(P^2) of double points of nodal genus g curves, the plane model approach collapses for g > 10.

Random Points. In our program, which picks plane models at random from an Zariski open subspace of M_g, we start by picking the nodes. However, over a small field F_q it is not a good idea to pick points individually, because there might be simply too few: |P^2(F_q )| = 1+q +q^2. What we should do is to pick a collection Γ of δ points in P^2(F̄_q ) that is defined over F_q . General points in P^2 satisfy the minimal resolution condition, that is, they have expected Betti numbers. This follows from the Hilbert-Burch theorem [12, Theorem 20.15]. If the ideal of such Γ has generators in minimal degree k, then (_{k+1} 2 ) ≤ δ < (_{k+2} 2 ) , which gives δ = (_{k+1} 2 ) + ϵ with 0 ≤ ϵ ≤ k. Thus k = ⌈(−3 + √ 9 + 8δ)/2⌉. The Betti table is one of the following two tables:

2ϵ ≤ k :

0 1 - - 1 - - - . . . . . . . . k − 2 - - - k − 1 - k + 1 − ϵ k − 2ϵ k - - ϵ

2ϵ ≥ k :

0 1 - - 1 - - - . . . . . . . . k − 2 - - - k − 1 - k + 1 − ϵ - k - 2ϵ − k ϵ So we can specify a collection Γ of δ points by picking the Hilbert-Burch matrix of their resolution; see [12, Thm 20.15]. This is a matrix with linear and quadratic entries only, whose minors of size ϵ (k − ϵ if 2ϵ ≤ k) generate the homogeneous ideal of Γ .

i1 : randomPlanePoints = (delta,R) -> (
          k:=ceiling((-3+sqrt(9.0+8*delta))/2);
          eps:=delta-binomial(k+1,2);
          if k-2*eps>=0
          then minors(k-eps,
               random(R^(k+1-eps),R^{k-2*eps:-1,eps:-2}))
          else minors(eps,
               random(R^{k+1-eps:0,2*eps-k:-1},R^{eps:-2})));

In unlucky cases these points might be infinitesimally near.

i2 : distinctPoints = (J) -> (
          singJ:=minors(2,jacobian J)+J;
          codim singJ == 3);

The procedure that returns the ideal of a random nodal curve is then straightforward:

i3 : randomNodalCurve = method();

i4 : randomNodalCurve (ZZ,ZZ,Ring) := (d,g,R) -> (
          delta:=binomial(d-1,2)-g;
          K:=coefficientRing R;
          if (delta==0)
          then ( --no double points
               ideal random(R^1,R^{-d}))
          else ( --delta double points
               Ip:=randomPlanePoints(delta,R);
               --choose the curve
               Ip2:=saturate Ip^2;
               ideal (gens Ip2 * random(source gens Ip2, R^{-d}))));

i5 : isNodalCurve = (I) -> (
          singI:=ideal jacobian I +I;delta:=degree singI;
          d:=degree I;g:=binomial(d-1,2)-delta;
          {distinctPoints(singI),delta,g});

We next ask if we indeed get in this way points in a parameter space that dominates M_g for g ≤ 10. Let Hilb_{(d,g)}(P^2) denote the Hilbert scheme of nodal plane curves of degree d and geometric genus g. Our construction starts from a random element in Hilb_δ(P^2) and picks a random curve in the corresponding fiber of Hilb_{(d,g)}(P^2) → Hilb_δ (P^2):

Hilb_{(d,g)}(P^2) � � M_g

Hilb_δ(P^2)

.

So the question is whether Hilb_{(d,g)}(P^2) dominates Hilb_δ (P^2). A naive dimension count suggests that this should be true: the dimension of our parameter space is given by 2δ + s, which is 3(g − 1) + ρ + dim PGL(3), as it should be. To conclude this there is more to verify: it could be that the nodal models of general curves have double points in special position, while all curve constructed above lie over a subvariety of M_g . One way to exclude this is to prove that the variety G(g, d, 2) over M_g , whose fiber over a curve C̃ ∈ M_g is G^2_d(C̃) = {g^2_d’s}, is irreducible or, to put it differently, that the Severi Conjecture holds:

Theorem 1.1 (Harris [20]). The space of nodal degree d genus g curves in P^2 is irreducible.

Another much easier proof for the few (d, g) we are interested in is to establish that our parameter space M of the construction is smooth of expected dimension at our random point p ∈ M , as in [1]. Consider the following diagram:

H = Hilb_{(d,g)}(P^2)/ Aut(P^2)^π−→ M_g .

For a given curve C̃ ∈ M_g , the inverse image π^{−1}(C̃) consists of the variety W^2_d (C̃) ⊂ Pic^d(C̃). Moreover the choice of a divisor L ∈ W^2_d (C̃) is equivalent to the choice of p ∈ M , modulo Aut(P^2): indeed p determines a morphism ν : C̃ −→ C ⊂ P^2 and a line bundle L = ν^{−1}O_P2(1), where C̃ is the normalization of the (nodal) curve C. Therefore M is smooth of expected dimension 3(g−1)+ρ+dim PGL(3) at p ∈ M if and only if W^2_d (C̃) is smooth of expected dimension ρ in L. This is well known to be equivalent to the injectivity of the multiplication map μ_L

H^0(L) ⊗ H^0(K_{C̃} ⊗ L^{−1})^{μL}−→ H^0(K_{C̃} ),

which can be easily checked in our cases, see [2, p. 189]. In our cases μ_L can be rewritten as

H^0(O_P2(1)) ⊗ H^0(I_Γ (d − 4))^{μL}−→ H^0(I_Γ (d − 3)).

So we need two conditions:

(1) H^0(I_Γ (d − 5)) = 0;

(2) there are no linear relations among the generators of H^0(I_Γ (d − 4)) of degree d − 3.

We proceed case by case. For genus g ≤ 5 this is clear, since H^0(I_Γ (d−4)) = 0 for g = 2, 3 and dim H^0(I_Γ (d − 3)) = 1 for g = 4, 5. For g = 6 we have dim H^0(I_Γ (d − 3)) = dim H^0(I_Γ (2)) = 2 and the Betti numbers of Γ

1 - - - 2 - - - 1

shows there are no relations with linear coefficients in H^0(I_Γ (2)). For 7 ≤ g ≤ 10 the method is similar: everything is clear once the Betti table of resolution of the set of nodal points Γ is computed. As a further example we do here the case g = 10: we see that dim H^0(I_Γ (d − 3)) = dim H^0(I_Γ (5)) = 3 and the Betti numbers of Γ are 1 - - - - - - - - - - - - 3 - - 1 3

from which it is clear that there are no linear relations between the quintic generators of I_Γ .

1.2 Space Models and Hartshorne-Rao Modules

The Case of Genus g = 11. In this case we have ρ(11, 12, 3) = 3. Hence every general curve of genus 11 has a space model of degree 12. Moreover for a general curve the general space model of this degree is linearly normal, because ρ(11, 13, 4) = −1 takes a smaller value. If moreover such a curve C ⊂ P^3 has maximal rank, i.e., for each m ∈ Z the map

H^0(P^3, O(m)) → H^0(C, O_C (m))

has maximal rank, then the Hartshorne-Rao module M, defined as M = H^1_∗ (I_C ) = ⊕_mH^1(P^3, I_C (m)), has Hilbert function with values (0, 0, 4, 6, 3, 0, 0, . . . ). For readers who want to know more about the Hartshorne-Rao module, we refer to the pleasant treatment in [24].

Since being of maximal rank is an open condition, we will try a construction of maximal rank curves. Consider the vector bundle G on P^3 associated to the first syzygy module of I_C :

(1) 0 ← I_C ← ⊕_iO(−a_i) ← G ← 0

In this set-up H^2_∗ (G) = H^1_∗ (I_C ). Thus G is, up to direct sum of line bundles, the sheafified second syzygy module of M; see e.g., [10, Prop. 1.5].

The expected Betti numbers of M are

4 10 3 - - - - 8 2 - - - - 6 3

Thus the F-dual M^∗ = Hom_F(M, F) is presented as F[x_0 , . . . , x_3]-module by a 3 × 8 matrix with linear and quadratic entries, and a general such matrix will give a general module (if the construction works, i.e., if the desired space of modules is non-empty), because all conditions we impose are semi-continuous and open. Thus M depends on

dim G (6, 3h^0 O(1)) + dim G (2, 3h^0 O(2) − 6h^0O(1)) − dim SL(3) = 36

parameters.

Assuming that C has minimal possible syzygies:

1 - - - - - - - - - - - - - - - - 6 2 - - - 6 3

we obtain, by dualizing the sequence (1), the following exact sequence G^∗←6O(5) ← O ← 0. If everything is as expected, i.e., the general curve is of maximal rank and its syzygies have minimal possible Betti numbers, then the entries of the right hand matrix are homogeneous polynomials that generate I_C . We will compute I_C by determining ker(ϕ: 6O(5) → G^∗). Comparing with the syzygies of M we obtain the following isomorphism

G^∗ ∼= ker(2O(6)⊕6O(7) → 3O(8)) ∼= image(3O(4)⊕8O(5) → 2O(6)⊕6O(7)).

and G^∗ ← 6O(5) factors over G^∗ ← 8O(5)⊕3O(4). A general ϕ ∈ Hom(6O(5), G^∗) gives a point in G (6, 8) and the Hilbert scheme of desired curves would have dimension 36 + 12 = 48 = 4 · 12 = 30 + 3 + 15 as expected, c.f. [19].

Therefore the computation for obtaining a random space curve of genus 11 is done as follows:

i6 : randomGenus11Curve = (R) -> (
          correctCodimAndDegree:=false;
          while not correctCodimAndDegree do (
               Mt=coker random(R^{3:8},R^{6:7,2:6});
               M=coker (transpose (res Mt).dd_4);
               Gt:=transpose (res M).dd_3;
               I:=ideal syz (Gt*random(source Gt,R^{6:5}));
               correctCodimAndDegree=(codim I==2 and degree I==12););
          I);

In general for these problems there is rarely an a priori reason why such a construction for general choices will give a smooth curve. Kleiman’s global generation condition [23] is much too strong a hypothesis for many interesting examples. But it is easy to check an example over a finite field with a computer:

i7 : isSmoothSpaceCurve = (I) -> (
          --I generates the ideal sheaf of a pure codim 2 scheme in P3
          singI:=I+minors(2,jacobian I);
          codim singI==4);

Hence by semi-continuity this is true over Q and the desired unirationality of G(11, 12, 3)/M_{11} holds for all fields, except possibly for those whose ground field has characteristic is in some finite set.

A calculation of an example over the integers would bound the number of exceptional characteristics, which then can be ruled out case by case, or by considering sufficiently many integer examples.

As in case of nodal curves, to prove unirationality of M_{11} by computer aided computations we have to show the injectivity of

H^0(L) ⊗ H^0(K_C ⊗ L^{−1})^{μL}−→ H^0(K_C ),

where L is the restriction of O_P3(1) to the curve C ⊂ P^3. The following few lines do the job:

i8 : K=ZZ/101;

i9 : R=K[x_0..x_3];

i10 : C=randomGenus11Curve(R);

o10 : Ideal of R

i11 : isSmoothSpaceCurve(C)

o11 = true

i12 : Omega=prune Ext^2(coker gens C,R^{-4});

i13 : betti Omega

o13 = relations : total: 5 10
                     -1: 2 .
                      0: 3 10

We see that there are no linear relations among the two generators of H^0_∗ (Ω_C ) in degree -1.

Betti Numbers for Genus g = 12, 13, 14, 15. The approach in these cases is similar to g = 11. We choose here d = g, so ρ(g, g, 3) = g − 12 ≥ 0. Under the maximal rank assumption the corresponding space curve has a Hartshorne-Rao module whose Hilbert function takes values (0, 0, g − 9, 2g − 19, 3g − 34, 0, . . .) in case g = 12, 13 and (0, 0, g − 9, 2g − 19, 3g − 34, 4g − 55, 0, . . .) in case g = 14, 15. Expected syzygies of M have Betti tables:

g = 12 :

3 7 - - - - - 10 5 - - - - 3 2 g = 13 :

4 9 1 - - - - 6 - - - - 6 13 5

g = 14 :

5 11 2 - - - - 3 - - - - 13 17 4 - - - - 1 g = 15 :

6 13 3 - - - - 3 - - - - 8 3 - - - - 9 5

Comparing with the expected syzygies of C

g = 12 :

1 - - - - - - - - - - - - - - - - 7 5 - - - 3 2 g = 13 :

1 - - - - - - - - - - - - - - - - 3 - - - 6 13 5

g = 14 :

1 - - - - - - - - - - - - - - - - - - - - 13 17 4 - - - 1 g = 15 :

1 - - - - - - - - - - - - - - - - - - - - 8 3 - - - 9 5

we see that given M the choice of a curve corresponds to a point in G (7, 10) or G (3, 6) for g = 12, 13 respectively, while for g = 14, 15 everything is determined by the Hartshorne-Rao module. For g = 12 Kleiman’s result guarantees smoothness for general choices, in contrast to the more difficult cases g = 14, 15. So the construction of M is the crucial step.

Construction of Hartshorne-Rao Modules. In case g = 12 the construction of M is straightforward. It is presented by a sufficiently general matrix of linear forms: 0 ← M ← 3S(−2) ← 7S(−3). The procedure for obtaining a random genus 12 curve is:

i14 : randomGenus12Curve = (R) -> (
           correctCodimAndDegree:=false;
           while not correctCodimAndDegree do (
                M:=coker random(R^{3:-2},R^{7:-3});
                Gt:=transpose (res M).dd_3;
                I:=ideal syz (Gt*random(source Gt,R^{7:5}));
                correctCodimAndDegree=(codim I==2 and degree I==12););
           I);

In case g = 13 we have to make sure that M has a second linear syzygy. Consider the end of the Koszul complex: 6R(−2)^κ← 4R(−3) ← R(−1) ← 0. Any product of a general map 4R(−2)^α← 6R(−2) with the Koszul matrix κ yields 4R(−2) ← 4R(−3) with a linear syzygy, and concatenated with a general map 4R(−2)^β← 5R(−3) gives a presentation matrix of a module M of desired type: 0 ← M ← 4R(−2) ← 4R(−3) ⊕ 5R(−3). The procedure for obtaining a random genus 13 curve is:

i15 : randomGenus13Curve = (R) -> (
           kappa:=koszul(3,vars R);
           correctCodimAndDegree:=false;
           while not correctCodimAndDegree do (
                test:=false;while test==false do (
                     alpha:=random(R^{4:-2},R^{6:-2});
                     beta:=random(R^{4:-2},R^{5:-3});
                     M:=coker(alpha*kappa|beta);
                     test=(codim M==4 and degree M==16););
                Gt:=transpose (res M).dd_3;
                --up to change of basis we can reduce phi to this form
                phi:=random(R^6,R^3)++id_(R^6);
                I:=ideal syz(Gt_{1..12}*phi);
                correctCodimAndDegree=(codim I==2 and degree I==13););
           I);

The case of genus g = 14 is about a magnitude more difficult. To start with, we can achieve two second linear syzygies by the same method as in the case g = 13. A general matrix 5R(−2)^α← 12R(−2) composed with 12R(−2)^{κ⊕κ}← 8R(−3) yields the first component of

5R(−2) ← (8 + 3)R(−3).

For a general choice of the second component 5R(−2)^β← 3R(−3) the cokernel will be a module with Hilbert function (0, 0, 5, 9, 8, 0, 0, . . .) and syzygies

5 11 2 - - - - 2 - - - - 17 23 8 - - - - -

What we want is to find α and β so that dim M_5 = 1 and dim Tor^R_2 (M, F)_5 = 3. Taking into account that we ensured dim Tor^R_2 (M, F)_4 = 2 this amounts to asking that the 100 × 102 matrix m(α, β) obtained from

[0 ← 5R(−2)_5 ← 11R(−3)_5 ← 2R(−4)_5 ← 0] ∼= [0 ← 100F^{m(α,β)}← 102F ← 0] drops rank by 1. We do not know a systematic approach to produce such m(α, β)’s. However, we can find such matrices in a probabilistic way. In the space of the matrices m(α, β), those which drop rank by 1 have expected codimension 3. Hence over a finite field F = F_q we expect to find the desired modules M with a probability of 1/q^3. The code to detect bad modules is rather fast.

i16 : testModulesForGenus14Curves = (N,p) ->(
           x := local x;
           R := ZZ/p[x_0..x_3];
           i:=0;j:=0;
           kappa:=koszul(3,vars R);
           kappakappa:=kappa++kappa;
           utime:=timing while (i<N) do (
                test:=false;
                alpha:=random(R^{5:-2},R^{12:-2});
                beta:=random(R^{5:-2},R^{3:-3});
                M:=coker (alpha*kappakappa|beta);
                fM:=res (M,DegreeLimit =>3);
                if (tally degrees fM_2)_{5}==3 then (
                     --further checks to pick up the right module
                     test=(tally degrees fM_2)_{4}==2 and
                     codim M==4 and degree M==23;);
                i=i+1;if test==true then (j=j+1;););
           timeForNModules:=utime#0; numberOfGoodModules:=j;
           {timeForNModules,numberOfGoodModules});

i17 : testModulesForGenus14Curves(1000,5)

o17 = {41.02, 10}

o17 : List

For timing tests we used a Pentium2 400Mhz with 128Mb of memory running GNU Linux. On such a machine examples can be tested at a rate of 0.04 seconds per example. Hence an approximate estimation of the CPU-time required to find a good example is q^3 · 0.04 seconds. Comparing this with the time to verify smoothness, which is about 12 seconds for an example of this degree, we see that up to |F_q | = q ≤ 13 we can expect to obtain examples within few minutes. Actually the computations for q = 2 and q = 3 take longer than for q = 5 on average, because examples of “good modules” tend to give singular curves more often. Here is a table of statistics which summarizes the situation.

q 2 3 5 7 11 13 smooth curves 100 100 100 100 100 100 1-nodal curves 75 53 31 16 10 8 reduced more singular 1012 142 24 11 2 0 non reduced curves 295 7 0 0 0 0 total number of curves 1482 302 155 127 112 108 percentage of smooth curves 6.7% 33% 65% 79% 89% 93% approx. time (in seconds) 7400 3100 2700 3400 6500 9500

The procedure for obtaining a random genus 14 curve is

i18 : randomGenus14Curve = (R) -> (
           kappa:=koszul(3,vars R);
           kappakappa:=kappa++kappa;
           correctCodimAndDegree:=false;
           count:=0;while not correctCodimAndDegree do (
                test:=false;
                t:=timing while test==false do (
                     alpha=random(R^{5:-2},R^{12:-2});
                     beta=random(R^{5:-2},R^{3:-3});
                     M:=coker (alpha*kappakappa|beta);
                     fM:=res (M,DegreeLimit =>3);
                     if (tally degrees fM_2)_{5}==3 then (
                          --further checks to pick up the right module
                          test=(tally degrees fM_2)_{4}==2 and
                          codim M==4 and degree M==23;);
                     count=count+1;);
                Gt:=transpose (res M).dd_3;
                I:=ideal syz (Gt_{5..17});
                correctCodimAndDegree=(codim I==2 and degree I==14););
           <<" -- "<<t#0<<" seconds used for ";
           <<count<<" modules"<<endl;
           I);

For g = 15 we do not know a method along these lines that would give examples over small fields.

Counting Parameters. For genus g = 12 clearly the module M depends on dim G (7, 3 · h^0O(1)) − dim SL(3) = 7 · 5 − 8 = 36 parameters, and the family of curves has dimension 36 + dim G (7, 10) = 48 = 4 · 12 = 33 + 0 + 15, as expected.

For genus g = 13 and 14 the parameter count is more difficult. Let us make a careful parameter count for genus g = 14; the case g = 13 is similar and easier. The choice of α corresponds to a point in G (5, 12). Then β corresponds to a point G (3, B_α ) where B_α = U ⊗ R_1/<α> where U denotes the universal subbundle on G (5, 12) and <α> the subspace generated by the 8 columns of α ∘ (κ ⊕ κ). So dim B_α = 20 − 8 = 12 and G (3, B_α ) → G (5, 12) is a Grassmannian bundle with fiber dimension 27 and total dimension 62. In this space the scheme of good modules has codimension 3, so we get a 59 dimensional family. This is larger than the expected dimension 56 = 4 · 14 = 39 + 2 + 15 of the Hilbert scheme, c.f. [19]. Indeed the construction gives a curve together with a basis of Tor^R_2 (M, F)_4 . Subtracting the dimension of the group of the projective coordinate changes we arrive at the desired dimension 59 − 3 = 56.

The unirationality of M_{12} and M_{13} can be proved by computer as in case M_{11}, while in case g = 14 we don’t know the unirationality of the parameter space of the modules M with dim M_5 = 1 and dim Tor^R_2 (M, F)_5 = 3.

2 Comparing Green’s Conjecture for Curves and Points

2.1 Syzygies of Canonical Curves

One of the most outstanding conjectures about free resolutions is Green’s prediction for the syzygies of canonical curves.

A canonical curve C ⊂ P^{g−1}, i.e., a linearly normal curve with O_C (1) ≡ ω_C , the canonical line bundle, is projectively normal by a result of Max Noether, and hence has a Gorenstein homogeneous coordinate ring and is 3-regular.

Therefore the Betti numbers of the free resolution of a canonical curve are symmetric, that is, β_{j,j+1} = β_{g−2−j,g−j} , and essentially only two rows of Betti numbers occur. The situation is summarized in the following table.

• 1 - - - - - - - - - - - - β_{1,2} β_{2,3} · · · β_{p,p+1} · · · · · · β_{p,p+2} - - - - - - - - β_{p,p+2} · · · · · · β_{p,p+1} · · · β_{2,3} β_{1,2} - - - - - - - - - - - - 1

The first p such that β_{p,p+2} ≠ 0 is conjecturally precisely the Clifford index of the curve.

Conjecture 2.1 (Green [16]). Let C be a smooth canonical curve over C . Then β_{p,p+2} ≠ 0 if and only if ∃L ∈ Pic^d(C) with h^0(C, L), h^1(C, L) ≥ 2 and cliff(L) := d − 2(h^0(C, L)) − 1) ≤ p. In particular, β_{j,j+2} = 0 for j ≤ ⌊^{g−3}_2 ⌋ for a general curve of genus g.

The “if” part is proved by Green and Lazarsfeld in [18] and holds for arbitrary ground fields. For some partial results see [35,31,30,4,5,22,25]. The conjecture is known to be false for some (algebraically closed) fields of finite characteristic, e.g., genus g = 7 and characteristic char F = 2; see [29].

2.2 Coble Self-Dual Sets of Points

The free resolution of a hyperplane section of a Cohen-Macaulay ring has the same Betti numbers. Thus we may ask for a geometric interpretation of the syzygies of 2g − 2 points in P^{g−2} (hyperplane section of a canonical curve), or syzygies of a graded Artinian Gorenstein algebra with Hilbert function (1, g − 2, g − 2, 1, 0, . . .) (twice a hyperplane section). Any collection of 2g − 2 points obtained as a hyperplane section of a canonical curve is special in the sense that it imposes only 2g − 3 conditions on quadrics. An equivalent condition for points in linearly uniform position is that they are Coble (or Gale) self-dual; see [14]. Thus if we distribute the 2g − 2 points into two collections each of g −1 points, with, say, the first consisting of the coordinate points and the second corresponding to the rows of a (g − 1) × (g − 1) matrix

A = (a_{ij} ), then A can be chosen to be an orthogonal matrix, i.e., A^tA = 1; see [14].

To see what the analogue of Green’s Conjecture for the general curve means for orthogonal matrices we recall a result of [28].

Set n = g − 2. We identify the homogeneous coordinate ring of P^n with the ring S = F[∂_0 , . . . , ∂_n] of differential operators with constant coefficients, ∂_i =^∂_{∂xi} . The ring S acts on F[x_0 , . . . , x_n] by differentiation. The annihilator of q = x^2_0 + . . . + x^2_n is a homogeneous ideal J ⊂ S such that S/J is a graded Artinian Gorenstein ring with Hilbert function (1, n + 1, 1) and socle induced by q, see [27], [12, Section 21.2 and related exercise 21.7]. The syzygy numbers of S/J are

1 - - - - - - -^n_{n+2} (_{n+3} 2 ) · · ·^{p(n+1−p)} n+2 (_{n+3} p+1 ) · · ·^n_{n+2} (_{n+3} n+1 ) - - - - - - - 1

A collection H_0, . . . , H_n of hyperplanes in P^n is said to form a polar simplex to q if and only if the collection Γ = {p_0, . . . , p_n} ⊂ P̌^n of the corresponding points in the dual space has its homogeneous ideal I_Γ ⊂ S contained in J.

In particular the set Λ consisting of the coordinate points correspond to a polar simplex, because ∂_i∂_j annihilates q for i ≠ j.

For any polar collection of points Γ the free resolution S_Γ is a subcomplex of the resolution S_{S/J} . Green’s conjecture for the generic curve of genus g = n + 2 would imply:

Conjecture 2.2. For a general Γ and the given Λ the corresponding Torgroups

Tor^S_k (S/I_Γ , F)_{k+1} ∩ Tor^S_k (S/I_Λ, F)_{k+1} ⊂ Tor^S_k (S/J, F)_{k+1}

intersect transversally.

Proof. A zero-dimensional non-degenerate scheme Γ ⊂ P^n of degree n+1 has syzygies 1 - - - - - - (_{n+1} 2 ) · · · k (_{n+1} k+1 ) · · · n Since both Tor groups are contained in Tor^S_k (S/J, F)_{k+1} , the claim is equivalent to saying that for a general polar simplex Γ the expected dimension of their intersection is dim Tor^S_k (S/_Γ , F)_{k+1} + dim Tor^S_k (S/_Γ , F)_{k+1} − dim Tor^S_k (S/J, F)_{k+1} , which is 2k ( g − 1 k + 1 ) − k(g − 1 − k) g ( g + 1 k + 1 ) .

On the other hand, I_{Γ ∪Λ} = I_Γ ∩ I_Λ, hence

Tor^S_k (S/I_Γ , F)_{k+1} ∩ Tor^S_k (S/I_Λ, F)_{k+1} = Tor^S_k (S/I_{Γ ∪Λ}, F)_{k+1} ,

and Green’s conjecture would imply

dim Tor^S_k (S/I_{Γ ∪Λ}, F)_{k+1} = β_{k,k+1}(Γ ∪ Λ) = k ( g − 2 k + 1 ) − (g − 1 − k) ( g − 2 k − 2 ) . Now a calculation shows that the two dimensions above are equal.

The family of all polar simplices V is dominated by the family defined by the ideal of 2 × 2 minors of the matrix ( ∂_0 . . . ∂_i . . . ∂_n ∑ j b_{0j} ∂_j . . . ∑ j b_{ij} ∂_j . . . ∑ j b_{nj} ∂_j )

depending on a symmetric matrix B = (b_{ij} ), i.e., b_{ij} = b_{ji} as parameters. For B a general diagonal matrix we get Λ together with a specific element in Tor^S_n (S/I_Λ, F)_{n+1} .

2.3 Comparison and Probes

One of the peculiar consequences of Green’s conjecture for odd genus g = 2k + 1 is that, if β_{k,k+1} = β_{k−1,k+1} ≠ 0, then the curve C lies in the closure of the locus of k + 1-gonal curve. Any k + 1-gonal curve lies on a rational normal scroll X of codimension k that satisfies β_{k,k+1}(X) = k. Hence β_{k,k+1}(C) ≠ 0 ⇒ β_{k,k+1}(C) ≥ k We may ask whether a result like this is true for the union of two polar simplices Λ ∪ Γ ⊂ P^{2k−1}. Define

D̃ = {Γ ∈ V |Γ ∪ Λ is syzygy special}

where, as above, V denotes the variety of polar simplices and Λ denotes the coordinate simplex. If D̃ is a proper subvariety, then it is a divisor, because β_{k,k+1}(Γ ∪ Λ) = β_{k−1,k+1}(Γ ∪ Λ). Conjecture 2.3. The subscheme D̃ ⊂ V is an irreducible divisor, for g = n + 2 = 2k + 1 ∈ {7, 9, 11}. The value of β_{k,k+1} on a general point of D is 3, 2, 1 respectively.

We can prove this for g = 7 by computer algebra. For g = 9 and g = 11 a proof is computationally out of reach with our methods, but we can get some evidence from examples over finite fields.

Evidence. Since D̃ is a divisor, we expect that if we pick symmetric matrices B over F_q at random, we will hit points on every component of D̃ at a probability of 1/q. For a general point on D̃ the corresponding Coble selfdual set of points will have the generic number of extra syzygies of that component. Points with even more syzygies will occur in higher codimension, hence only with a probability of 1/q^2. Some evidence for irreducibility can be obtained from the Weil formulas: for sufficiently large q we should see points on D̃ with probability Cq^{−1} +O(q^{− 32} ), where C is the number of components.

The following tables give for small fields F_q the number s_i of examples with i extra syzygies in a test of 1000 examples for g = 9 and 100 examples for g = 11. The number s_{tot} = ∑ i>0 s_i is the total number of examples with extra syzygies. Genus g = 9:

q 1000/q s_{tot} s_1 s_2 s_3 s_4 2 500 925 0 130 0 63 3 333 782 0 273 0 33 4 250 521 0 279 0 99 5 200 350 0 217 0 74 7 143 197 0 144 0 36 8 125 199 0 147 0 43 9 111 218 0 98 0 0 11 91 118 0 102 0 15 13 77 90 0 79 0 10 16 62 72 0 68 0 4 17 59 76 0 69 0 6

Genus g = 11:

q 100/q s_{tot} s_1 s_2 s_3 s_4 7 14.3 16 14 0 0 0 17 5.9 7 7 0 0 0

In view of these numbers, it is more likely that the set D̃ of syzygy special Coble points is irreducible than that it is reducible. For a more precise statement we refer to [6].

A test of Green’s Conjecture for Curves. In view of 2.3 it seems plausible that for a general curve of odd genus g ≥ 11 with β_{k,k+1}(C) ≠ 0 the value might be β_{k,k+1} = 1 contradicting Green’s conjecture. It is clear that the syzygy exceptional locus has codimension 1 in M_g for odd genus, if it is proper, i.e., if Green’s conjecture holds for the general curve of that genus. So picking points at random we might be able to find such curve over a finite field F_q with probability 1/q, roughly.

Writing code that does this is straightforward. One makes a loop that picks up randomly a curve, computes its canonical image, and resolves its ideal, counting the possible values β_{k,k+1} until a certain amount of special curves is reached. The result for 10 special curves in F_7 is as predicted:

total special possible values of β_{k,k+1} g seconds curves curves ≤ 2 3 4 5 6 ≥ 7 7 148 75 10 0 10 0 0 0 0 9 253 58 10 0 0 9 0 0 1 11 25640 60 10 0 0 0 9 0 1

(The test for genus 9 and 11 used about 70 and 120 megabytes of memory, respectively.)

So Green’s conjecture passed the test for g = 9, 11. Shortly after the first author tried this test for the first time, a paper of Hirschowitz and Ramanan appeared proving this in general:

Theorem 2.4 ([22]). If the general curve of odd genus g = 2k + 1 satisfies Green’s conjecture then the syzygy special curves lie on the divisor D = {C ∈ M_g |W^1_{k+1}(C) ≠ ∅}

The theorem gives strong evidence for the full Green’s conjecture in view of our study of Coble self-dual sets of points.

Our findings suggest that the variety of points arising as hyperplane sections of smooth canonical curves has the strange property that it intersects the divisor of syzygy special sets of points D̃ only in its singular locus.

The conjecture for general curves is known to us up to g ≤ 17, which is as far as a computer allows us to do a ribbon example; see [4].

3 Pfaffian Calabi-Yau Threefolds in P^6

Calabi-Yau 3-folds caught the attention of physicists because they can serve as the compact factor of the Kaluza-Klein model of spacetime in superstring theory. One of the remarkable things that grows out of the work in physics is the discovery of mirror symmetry, which associates to a family of Calabi-Yau 3-folds (M_λ), another family (W_μ) whose Hodge diamond is the mirror of the Hodge diamond of the original family.

Although there is an enormous amount of evidence at present, the existence of a mirror is still a hypothesis for general Calabi-Yau 3-folds. The thousands of cases where this was established all are close to toric geometry, where through the work of Batyrev and others [3,9] rigorous mirror constructions were given and parts of their conjectured properties proved.

From a commutative algebra point of view the examples studied so far are rather trivial, because nearly all are hypersurfaces or complete intersections on toric varieties, or zero loci of sections in homogeneous bundles on homogeneous spaces.

Of course only a few families of Calabi-Yau 3-folds should be of this kind. Perhaps the easiest examples beyond the toric/homogeneous range are Calabi-Yau 3-folds in P^6. Here examples can be obtained by the Pfaffian construction of Buchsbaum-Eisenbud [7] with vector bundles; see section 3.1 below. Indeed a recent theorem of Walter [36] says that any smooth Calabi- Yau in P^6 can be obtained in this way. In this section we report on our construction of such examples.

As is quite usual in this kind of problem, there is a range where the construction is still quite easy, e.g., for surfaces in P^4 the work in [10,26] shows that the construction of nearly all the 50 known families of smooth non-general type surfaces is straight forward and their Hilbert scheme component unirational. Only in very few known examples is the construction more difficult and the unirationality of the Hilbert scheme component an open problem.

The second author did the first “non-trivial” case of a construction of Calabi-Yau 3-folds in P^6. Although in the end the families turned out to be unirational, the approach utilized small finite field constructions as a research tool.

3.1 The Pfaffian Complex

Let F be a vector bundle of odd rank rk F = 2r + 1 on a projective manifold M, and let L be a line bundle. Let φ ∈ H^0(M, Λ^2F ⊗ L) be a section. We can think of φ as a skew-symmetric twisted homomorphism F^{∗ φ}−→ F ⊗ L. The r^{th} divided power of φ is a section φ^{(r)} =^1_{r!} (φ ∧ · · · ∧ φ) ∈ H^0(M, Λ^{2r}F ⊗ L^r). Wedge product with φ^{(r)} defines a morphism

F ⊗ L^ψ−→ Λ^{2r+1}F ⊗ L^{r+1} = det(F) ⊗ L^{r+1}.

The twisted image I = image(ψ) ⊗ det(F^∗) ⊗ L^{−r−1} ⊂ O_M is called the Pfaffian ideal of φ, because working locally with frames, it is given by the ideal generated by the 2r × 2r principle Pfaffians of the matrix describing φ. Let D denote the determinant line bundle det(F^∗).

Theorem 3.1 (Buchsbaum-Eisenbud [7]). With this notation 0 → D^2 ⊗ L^{−2r−1 ψt}−→ D ⊗ L^{−r−1} ⊗ F^{∗ φ}−→ F ⊗ D ⊗ L^{−r ψ}−→ O_M is a complex. X = V (I) ⊂ M has codimension ≤ 3 at every point, and in case equality holds (everywhere along X) then this complex is exact and resolves the structure sheaf O_X = O_M /I of the locally Gorenstein subscheme X.

We will apply this to construct Calabi-Yau 3-folds in P^6. In that case we want X to be smooth and det(F)^{−2} ⊗ L^{−2r−1} ∼= ω_P ∼= O(−7), so we may conclude that ω_X ∼= O_X . A result of Walter [36] for P^n guarantees the existence of a Pfaffian presentation in P^6 for every subcanonical embedded 3- fold. Moreover Walter’s choice of F ⊗D⊗L^{−r} for Calabi-Yau 3-folds X ⊂ P^6 is the sheafified first syzygy module H^1_∗ (I_X ) plus possibly a direct sum of line bundles (indeed H^2_∗ (I_X ) = 0 because of the Kodaira vanishing theorem). Under the maximal rank assumption for

H^0(P^6, O(m)) → H^0(X, O_X (m))

the Hartshorne-Rao module is zero for d = deg X ∈ {12, 13, 14} and an arithmetically Cohen-Macaulay X is readily found. For d ∈ {15, 16, 17, 18} the Hartshorne-Rao modules M have Hilbert functions with values (0, 0, 1, 0, . . .), (0, 0, 2, 1, 0, . . .), (0, 0, 3, 5, 0, . . .) and (0, 0, 4, 9, 0, . . .) respectively.

We do not discuss the cases d ≤ 16 further. The construction in those cases is obvious; see [34].

3.2 Analysis of the Hartshorne-Rao Module for Degree 17

Denote with F_1 the sheaf F ⊗D⊗L^{−r}. We try to construct F_1 as the sheafified first syzygy module of M. The construction of a module with the desired Hilbert function is straightforward. The cokernel of 3S(−2)^b←− 16S(−3) for a general matrix of linear forms has this property. However, for a general b and F_1 = ker(16O(−3)^b−→ 3O(−2)) the space of skew-symmetric maps Hom_{skew}(F^∗_1 (−7), F_1) is zero: M has syzygies

3 16 28 - - - - - - - 70 112 84 32 5

Any map φ: F^∗_1 (−7) → F_1 induces a map on the free resolutions:

0 F1

� 28O(−4)

� 70O(−6)

� 112O(−7) �

0 F^∗_1 (−7) � φ �

16O(−4) � φ_0 �

3O(−5) � φ_1 �

0 � Since φ_1 = 0 for degree reasons, φ = 0 as well, and Hom(F^∗_1 (−7), F_1) = 0 for a general module M.

What we need are special modules M that have extra syzygies

3 16 28 k - - - - - k 70 112 84 32 5

with k at least 3.

20 F-O. Schreyer and F. Tonoli

In a neighborhood of o ∈ Spec B, where denotes B the base space of a semi-universal deformation of M, the resolution above would lift to a complex over B[x_0, . . . , x_6] and in the lifted complex there is a k × k matrix Δ with entries in the maximal ideal o ⊂ B. By the principal ideal theorem we see that Betti numbers stay constant in a subvariety of codimension at most k^2. To check for second linear syzygies on a randomly chosen M is a computationally rather easy task. The following procedure tests the computer speed of this task.

i19 : testModulesForDeg17CY = (N,k,p) -> (
           x:=symbol x;R:=(ZZ/p)[x_0..x_6];
           numberOfGoodModules:=0;i:=0;
           usedTime:=timing while (i<N) do (
                b:=random(R^3,R^{16:-1});
                --we put SyzygyLimit=>60 because we expect
                --k<16 syzygies, so 16+28+k<=60
                fb:=res(coker b,
                     DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
                if rank fb_3>=k and (dim coker b)==0 then (
                     fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
                     if rank fb_4==0
                     then numberOfGoodModules=numberOfGoodModules+1;);
                i=i+1;);
           collectGarbage();
           timeForNModules:=usedTime#0;
           {timeForNModules,numberOfGoodModules});

Running this procedure we see that it takes not more than 0.64 seconds per example. Hence we can hope to find examples with k = 3 within a reasonable time for a very small field, say F_3 .

The first surprise is that examples with k extra syzygies are found much more often, as can be seen by looking at the second value output by the function testModulesForDeg17CY().

This is not only a “statistical” remark, in the sense that the result is confirmed by computing the semi-universal deformations of these modules. Indeed define M_k = {M | Tor^S_3 (M, F)_5 ≥ k} and consider a module M ∈ M_k : “generically” we obtain codim(M_k )_M = k instead of k^2 (and in fact one can diagonalize the matrix Δ over the algebraic closure F̄).

The procedure is straightforward but a bit long. First we pick up an example with k-extra syzygies.

i20 : randomModuleForDeg17CY = (k,R) -> (
           isGoodModule:=false;i:=0;
           while not isGoodModule do (
                b:=random(R^3,R^{16:-1});
                --we put SyzygyLimit=>60 because we expect
                --k<16 syzygies, so 16+28+k<=60
                fb:=res(coker b,
                     DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
                if rank fb_3>=k and (dim coker b)==0 then (
                     fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
                     if rank fb_4==0 then isGoodModule=true;);
                i=i+1;);
           <<" -- Trial n. " << i <<", k="<< rank fb_3 <<endl;

Needles in a Haystack: Special Varieties via Small Fields 21

b);

Notice that the previous function returns a presentation matrix b of M, and not M.

Next we compute the tangent codimension of M_k in the given example M = Coker b by computing the codimension of the space of the infinitesimal deformations of M that still give an element in M_k . Denote with b_i the maps in the linear strand of a minimal free resolution of M, and with b^′_2 the quadratic part in the second map of this resolution. Over B = F[ϵ]/ϵ^2 let b_1 + ϵf_1 be an infinitesimal deformation of b_1. Then f_1 lifts to a linear map f_2 : 28S(−4) → 16S(−3) determined by (b_1 +ϵf_1)∘(b_2 +ϵf_2) = 0, and f_2 to a map f_3⊕Δ: kS(−5) → 28S(−4)⊕kS(−5) determined by (b_2+b^′_2)∘ϵ(f_3⊕Δ) = 0. Therefore we can determine Δ as:

i21 : getDeltaForDeg17CY = (b,f1) -> (
           fb:=res(coker b, LengthLimit =>3);
           k:=numgens target fb.dd_3-28; --# of linear syzygies
           b1:=fb.dd_1;b2:=fb.dd_2_{0..27};b2’:=fb.dd_2_{28..28+k-1};
           b3:=fb.dd_3_{0..k-1}^{0..27};
           --the equation for f2 is b1*f2+f1*b2=0,
           --so f2 is a lift of (-f1*b2) through b1
           f2:=-(f1*b2)//b1;
           --the equation for A=(f3||Delta) is -f2*b3 = (b2|b2’) * A
           A:=(-f2*b3)//(b2l|b2’);
           Delta:=A^{28..28+k-1});

Now we just parametrize all possible maps f_1 : 16S(−3) → 3S(−2), compute their respective maps Δ, and find the codimension of the condition that Δ is the zero map:

i22 : codimInfDefModuleForDeg17CY = (b) -> (
           --we create a parameter ring for the matrices f1’s
           R:=ring b;K:=coefficientRing R;
           u:=symbol u;U:=K[u_0..u_(3*16*7-1)];
           i:=0;while i<3 do (
                <<endl<< " " << i+1 <<":" <<endl;
                j:=0;while j<16 do(
                     << " " << j+1 <<". "<<endl;
                     k:=0;while k<7 do (
                        l=16*7*i+7*j+k; --index parametrizing the f1’s
                        f1:=matrix(R,apply(3,m->apply(16,n->
                             if m==i and n==j then x_k else 0)));
                        Delta:=substitute(getDeltaForDeg17CY(b,f1),U);
                        if l==0 then (equations=u_l*Delta;) else (
                             equations=equations+u_l*Delta;);
                        k=k+1;);
                     collectGarbage(); --frees up memory in the stack
                     j=j+1;);
                i=i+1;);
           codim ideal equations);

The second surprise is that for F_1 = syz_1(M) we find

dim Hom_{skew}(F^∗_1 (−7), F_1) = k = dim Tor^S_3 (M, F)_5 .

Hom_{skew}(F^∗_1 (−7), F_1) is the vector space of skew-symmetric linear matrices φ such that b ∘ φ = 0. The following procedure gives a matrix of size

(_{16} 2 ) × dim Hom_{skew}(F^∗_1 (−7), F_1) whose i-th column gives the entries of a 16 × 16 skew-symmetric matrix inducing the i-th basis element of the vector space Hom_{skew}(F^∗_1 (−7), F_1).

i23 : skewSymMorphismsForDeg17CY = (b) -> (
           --we create a parameter ring for the morphisms:
           K:=coefficientRing ring b;
           u:=symbol u;U:=K[u_0..u_(binomial(16,2)-1)];
           --now we compute the equations for the u_i’s:
           UU:=U**ring b;
           equationsInUU:=flatten (substitute(b,UU)*
                substitute(genericSkewMatrix(U,u_0,16),UU));
           uu:=substitute(vars U,UU);
           equations:=substitute(
                diff(uu,transpose equationsInUU),ring b);
           syz(equations,DegreeLimit =>0));

A morphism parametrized by a column skewSymMorphism is then recovered by the following code.

i24 : getMorphismForDeg17CY = (SkewSymMorphism) -> (
           u:=symbol u;U:=K[u_0..u_(binomial(16,2)-1)];
           f=map(ring SkewSymMorphism,U,transpose SkewSymMorphism);
           f genericSkewMatrix(U,u_0,16));

Rank 1 Linear Syzygies of M. To understand this phenomenon we consider the multiplication tensor of M:

μ: M_2 ⊗ V → M_3

where V = H^0(P^6, O(1)).

Definition 3.2. A decomposable element of M_2 ⊗ V in the kernel of μ is called a rank 1 linear syzygy of M. The (projective) space of rank 1 syzygies is Y = (P^2 × P^6) ∩ P^{15} ⊂ P^{20}

where P^2 = P(M^∗_2 ), P^6 = P(V^∗) and P^{15} = P(ker(μ)^∗) inside the Segre space P((M_2 ⊗ V )^∗) ∼= P^{20}.

Proposition 1.5 of [17] says that, for dim M_2 ≤ j, the existence of a j^{th} linear syzygy implies dim Y ≥ j − 1. This is automatically satisfied for j = 3 in our case: dim Y ≥ 3 with equality expected.

The projection Y → P^2 has linear fibers, and the general fiber is a P^1. However, special fibers might have higher dimension. In terms of the presentation matrix b a special 2-dimensional fiber (defined over F) corresponds to a block b = 0 0 0 ∗ . . . 0 0 0 ∗ . . . l_1 l_2 l_3 ∗ . . . , where l_1, l_2, l_3 are linear forms, in the 3 × 16 presentation matrix of M. Such a block gives a 1 3 3 1 - - - - - - - - - - - -

subcomplex in the free resolution of M and an element s ∈ H^0(P^6, Λ^2F_1 ⊗ O(7)) since the syzygy matrix

0 −l_3 l_2 l_3 0 −l_1 −l_2 l_1 0

is skew.

This answers the questions posed by both surprises: we want a module M with at least k ≥ 3 special fibers and these satisfy h^0(P^6, Λ^2F_1 ⊗ O(7)) ≥ k, if the k sections are linearly independent. The condition for k special fibers is of expected codimension k in the parameter space G (16, 3h^0 (P^6, O(1)) of the presentation matrices. In a given point M the actual codimension can be readily computed by a first order deformations and that H^0(P^6, Λ^2F_1 ⊗O(7)) is k-dimensional, and spanned by the k sections corresponding to the k special fibers can be checked as well.

First we check that M has k distinct points in P(M^∗_2 ) where the multiplication map drops rank. (Note that this condition is likely to fail over small fields. However, the check is computationally easy).

i25 : checkBasePtsForDeg17CY = b -> (
           --firstly the number of linear syzygies
           fb:=res(coker b, DegreeLimit=>0, LengthLimit =>4);
           k:=#select(degrees source fb.dd_3,i->i=={3});
           --then the check
           a=symbol a;A=K[a_0..a_2];
           mult:=(id_(A^7)**vars A)*substitute(
                syz transpose jacobian b,A);
           basePts=ideal mingens minors(5,mult);
           codim basePts==2 and degree basePts==k and distinctPoints(
                basePts));

Next we check that H^0(P^6, Λ^2F_1 ⊗ O(7)) is k-dimensional, by looking at the numbers of columns of skewSymMorphismsForDeg17CY(b). Finally we do the computationally hard part of the check, which is to verify that the k special sections corresponding to the k special fibers of Y → P^2 span H^0(P^6, Λ^2F_1 ⊗ O(7)).

i26 : checkMorphismsForDeg17CY = (b,skewSymMorphisms) -> (
           --first the number of linear syzygies
           fb:=res(coker b, DegreeLimit=>0, LengthLimit =>4);
           k:=#select(degrees source fb.dd_3,i->i=={3});
           if (numgens source skewSymMorphisms)!=k then (
                error "the number of skew-sym morphisms is wrong";);
           --we parametrize the morphisms:
           R:=ring b;K:=coefficientRing R;
           w:=symbol w;W:=K[w_0..w_(k-1)];
           WW:=R**W;ww:=substitute(vars W,WW);

24 F-O. Schreyer and F. Tonoli

genericMorphism:=getMorphismForDeg17CY(
     substitute(skewSymMorphisms,WW)*transpose ww);
--we compute the scheme of the 3x3 morphisms:
equations:=mingens pfaffians(4,genericMorphism);
equations=diff(
     substitute(symmetricPower(2,vars R),WW),equations);
equations=saturate ideal flatten substitute(equations,W);
CorrectDimensionAndDegree:=(
     dim equations==1 and degree equations==k);
isNonDegenerate:=#select(
     (flatten degrees source gens equations),i->i==1)==0;
collectGarbage();
isOK:=CorrectDimensionAndDegree and isNonDegenerate;
if isOK then (
     --in this case we also look for a skew-morphism f
     --which is a linear combination of the special
     --morphisms with all coefficients nonzero.
     isGoodMorphism:=false;while isGoodMorphism==false do (
          evRandomMorphism:=random(K^1,K^k);
          itsIdeal:=ideal(
               vars W*substitute(syz evRandomMorphism,W));
          isGoodMorphism=isGorenstein(
               intersect(itsIdeal,equations));
          collectGarbage());
     f=map(R,WW,vars R|substitute(evRandomMorphism,R));
     randomMorphism:=f(genericMorphism);
     {isOK,randomMorphism}) else {isOK});

The code above is structured as follows. First we parametrize the skewsymmetric morphisms with new variables. The ideal of 4 × 4 Pfaffians is generated by forms of bidegree (2, 2) over P^6 × P^{k−1}. We are interested in points p ∈ P^{k−1} such that the whole fiber P^6 ×{p} is contained in the zero locus of the Pfaffian ideal. The next two lines produce the ideal of these points on P^{k−1}. Since we already know of k distinct points by the previous check, it suffices to establish that the set consists of collection of k spanning points. Finally, if this is the case, a further point, i.e., a further skew morphism, is a linear combination with all coefficients non-zero, if and only if the union with this point is a Gorenstein set of k + 1 points in P^{k−1}.

i27 : isGorenstein = (I) -> (
           codim I==length res I and rank (res I)_(length res I)==1);

It is clear that all 16 relations should take part in the desired skew homomorphism F^∗_1 (−7)^φ−→ F_1. Thus we need k ≥ 6 to have a chance for a Calabi-Yau. Since 3 · 5 < 16 it easy to guarantee 5 special fibers by suitable choice of the presentation matrix. So the condition k ≥ 6 is only of codimension k − 5 on this subspace, and we have a good chance to find a module of the desired type.

i28 : randomModule2ForDeg17CY = (k,R) -> (
           isGoodModule:=false;i:=0;
           while not isGoodModule do (
                b:=(random(R^1,R^{3:-1})++
                     random(R^1,R^{3:-1})++
                     random(R^1,R^{3:-1})|
                     matrix(R,{{1},{1},{1}})**random(R^1,R^{3:-1})|
                     random(R^3,R^1)**random(R^1,R^{3:-1})|

Needles in a Haystack: Special Varieties via Small Fields 25

          random(R^3,R^{1:-1}));
     --we put SyzygyLimit=>60 because we expect
     --k<16 syzygies, so 16+28+k<=60
     fb:=res(coker b,
          DegreeLimit =>0,SyzygyLimit=>60,LengthLimit =>3);
     if rank fb_3>=k and dim coker b==0 then (
          fb=res(coker b, DegreeLimit =>0,LengthLimit =>4);
          if rank fb_4==0 then isGoodModule=true;);
     i=i+1;);
<<" -- Trial n. " << i <<", k="<< rank fb_3 <<endl;
b);

Some modules M with k = 8, 9, 11 lead to smooth examples of Calabi-Yau 3-folds X of degree 17. To check the smoothness via the Jacobian criterion is computationally too heavy for a common computer today. For a way to speed up this computation considerably and to reduce the required amount of memory to a reasonable value (128MB), we refer to [34].

Since h^0(P^6, Λ^2F_1 ⊗ O(7)) = k and codim{M | Tor^S_3 (M, F)_5 ≥ k} = k all three families have the same dimension. In particular no family lies in the closure of another.

A deformation computation verifies h^1(X, T ) = h^1(X, Ω^2) = 23. Hence a computation of the Hodge numbers h^q (X, Ω^p) gives the diamond 1 0 0 0 1 0 1 23 23 1

0 1 0 0 0 1 Example 3.3. The following commands give an example of a Calabi-Yau 3-fold in P^6:

i29 : K=ZZ/13;

i30 : R=K[x_0..x_6];

i31 : time b=randomModule2ForDeg17CY(8,R);
     -- Trial n. 1757, k=8
     -- used 764.06 seconds

              3 16
o31 : Matrix R <--- R

i32 : betti res coker b

o32 = total: 3 16 36 78 112 84 32 5
          0: 3 16 28 8 . . . .
          1: . . 8 70 112 84 32 5

i33 : betti (skewSymMorphisms=skewSymMorphismsForDeg17CY b)

o33 = total: 120 8
         -1: 120 8

26 F-O. Schreyer and F. Tonoli We check whether the base points in M_0 are all distinct.

i34 : checkBasePtsForDeg17CY b
o34 = true

Now we check whether the k sections span the morphisms. If we get true then this is a good module.

i35 : finalTest=checkMorphismsForDeg17CY(b,skewSymMorphisms);
i36 : finalTest#0
o36 = true

We pick up a random morphism involving all k sections.

i37 : n=finalTest#1;
              16 16
o37 : Matrix R <--- R

If all the tests are okay, there should be a high degree syzygy.

i38 : betti (nn=syz n)
o38 = total: 16 4
          1: 16 3
          2: . .
          3: . 1
i39 : n2t=transpose submatrix(nn,{0..15},{3});
              1 16
o39 : Matrix R <--- R
i40 : b2:=syz b;
              16 36
o40 : Matrix R <--- R

Finally, compute the ideal of the Calabi-Yau 3-fold in P^6.

i41 : j:=ideal mingens ideal flatten(n2t*b2);
o41 : Ideal of R
i42 : degree j
o42 = 17
i43 : codim j
o43 = 3
i44 : betti res j
o44 = total: 1 20 75 113 84 32 5
          0: 1 . . . . . .
          1: . . . . . . .
          2: . . . . . . .
          3: . 12 5 . . . .
          4: . 8 70 113 84 32 5

3.3 Lift to Characteristic Zero

At this point we have constructed Calabi-Yau 3-folds X ⊂ P^6 over the finite field F_5 or F_7 . However, our main interest is the field of complex numbers C . The existence of a lift to characteristic zero follows by the following argument.

The set M_k = {M | Tor^S_3 (M, F)_5 ≥ k} has codimension at most k. A deformation calculation shows that at our special point M^{special} ∈ M (F_p ) the codimension is achieved and that M_k is smooth at this point. Thus taking a transversal slice defined over Z through this point we find a number field K and a prime p in its ring of integers O_K with O_K /p ∼= F_p such that M^{special} is the specialization of an O_{K,p}-valued point of M_k . Over the generic point of Spec O_{K,p} we obtain a K-valued point. From our computations with

checkBasePtsForDeg17CY() and checkMorphismsForDeg17CY(), which ex-

plained why h^0(P^6, Λ^2F^{special}_1 ⊗ O(7)) = k, it follows that

H^0(P^6_Z× Spec O_{K,p}, Λ^2F_1 ⊗ O(7))

is free of rank k over O_{K,p}. Hence φ^{special} extends to O_{K,p} as well, and by semi-continuity we obtain a smooth Calabi-Yau 3-fold defined over K ⊂ C .

Theorem 3.4 ([34]). The Hilbert scheme of smooth Calabi-Yau 3-folds of degree 17 in P^6 has at least 3 components. These three components are reduced and have dimension 23 + 48. The corresponding Calabi-Yau 3-folds differ in the number of quintic generators of their homogeneous ideals, which are 8, 9 and 11 respectively.

See [34] for more details.

Note that we do not give a bound on the degree [K : Q] of the number field, and certainly we are far away from a bound of its discriminant.

This leaves the question open whether these parameter spaces of Calabi- Yau 3-folds are unirational. Actually they are, as the geometric construction of modules M ∈ M_k in [34] shows.

A construction of one or several mirror families of these Calabi-Yau 3-folds is an open problem.

References

1. E. Arbarello and M. Cornalba: A few remarks about the variety of irreducible plane curve of given degree and genus. Ann. Sci. Ècole Norm. Sup(4), 16:467– 488, 1983.

2. E. Arbarello, M. Cornalba, P. Griffith, and J. Harris: Geometry of algebraic curves, vol I. Springer Grundlehren, 267:xvi+386, 1985.

3. V. Batyrev: Dual polyhedra and mirror symmetry for Calabi-Yau hypersurfaces in algebraic tori. J. Alg. Geom., 3:493–535, 1994.

4. D. Bayer and D. Eisenbud: Ribbons and their canonical embeddings. Trans.

Am. Math. Soc., 347,:719–756, 1995.

5. H.-C. v. Bothmer: Geometrische Syzygien kanonischer Kurven. Thesis Bayreuth, pages 1–123, 2000.

6. H.-C. v. Bothmer and F.-O. Schreyer: A quick and dirty irreduciblity test.

manuscript, to appear.

7. D. Buchsbaum and D. Eisenbud: Algebra structure on finite free resolutions and some structure theorem for ideals of codimension 3. Am. J. Math., 99:447–485, 1977.

8. M.-C. Chang and Z. Ran: Unirationality of the moduli space of curves genus 11, 13 (and 12). Invent. Math., 76:41–54, 1984.

9. D. Cox and S. Katz: Mirror symmetry and algebraic geometry. AMS, Mathematical Surveys and Monographs, 68:xxii+469, 1999.

10. W. Decker, L. Ein, and F.-O. Schreyer: Construction of surfaces in P^4. J. of Alg. Geom., 2:185–237, 1993.

11. W. Decker and F.-O. Schreyer: Non-general type surfaces in P^4 - Remarks on bounds and constructions. J. Symbolic Comp., 29:545–585, 2000.

12. D. Eisenbud: Commutative Algebra. With a view towards algebraic geometry.

Springer Graduate Texts in Mathematics, 150:xvi+785, 1995.

13. D. Eisenbud and J. Harris: The Kodaira dimension of the moduli space of curves of genus g ≥ 23. Invent. Math., 87:495–515, 1987.

14. D. Eisenbud and S. Popescu: Gale duality and free resolutions of ideals of points.. Invent. Math., 136:419–449, 1999.

15. G. Ellingsrud and C. Peskine: Sur les surfaces de P^4. Invent. Math., 95:1–11, 1989.

16. M. Green: Koszul cohomology and the geometry of projective varieties. J. Diff.

Geom., 19:125–171, 1984.

17. M. Green: The Eisenbud-Koh-Stillman conjecture on linear syzygies. Invent.

Math, 163:411–418, 1999.

18. M. Green and R. Lazarsfeld: Appendix to Koszul cohomology and the geometry of projective varieties. J. Diff. Geom., 19:168–171, 1984.

19. J. Harris: Curves in projective space. With collaboration of David Eisenbud.

Séminaire de Mathématiques Supériores, 85:Montreal, 138 pp., 1982. 20. J. Harris: On the Severi problem. Invent. Math., 84:445–461, 1986. 21. J. Harris and D. Mumford: On the Kodaira dimension of the moduli space of curves. With an appendix by William Fulton. Invent. Math., 67:23–88, 1982.

22. A. Hirschowitz and S. Ramanan: New evidence for Green’s conjecture on syzgyies of canonical curves. Ann. Sci. École Norm. Sup. (4), 31:145–152, 1998.

23. S. Kleiman: Geometry on Grassmannians and application to splitting bundles and smoothing cycles. Inst. Hautes Études Sci. Publ. Math. No., 36:281–287, 1969.

24. M. Martin-Deschamps and D. Perrin: Sur la classification des courbes gauches.

Asterisque, 184-185:1–208, 1990.

25. S. Mukai: Curves and symmetric spaces, I.. Amer. J. Math., 117:1627–1644, 1995.

26. S. Popescu: Some examples of smooth non general type surfaces in P^4. Thesis Saarbrücken, pages ii+123, 1994.

27. K. Ranestad and F.-O. Schreyer: Varieties of sums of powers. J. Reine Angew.

Math., 525:147–182, 2000.

28. K. Ranestad and F.-O. Schreyer: On the variety of polar simplices. manuscript, to be completed.

29. F.-O. Schreyer: Syzygies of canonical curves and special linear series. Math.

Ann., 275:105–137, 1986.

30. F.-O. Schreyer: Green’s conjecture for p-gonal curves of large genus. Algebraic curves and projective geometry (Trento 1988). Lecture Notes in Math., 1396:254–260, 1989.

31. F.-O. Schreyer: A standard basis approach to syzygies of canonical curves. J.

Reine Angew. Math., 421:83–123, 1991.

32. F.-O. Schreyer: Small fields in constructive algebraic geometry. In Moduli of vector bundles (Sanda, 1994; Kyoto, 1994), pages 221–228. Dekker, New York, 1996.

33. E. Sernesi: L’ unirazionalità della varieta dei moduli delle curve di genere dodici. Ann. Scuola Norm. Sup. Pisa Cl. Sci. (4), 8:405–439, 1981.

34. F. Tonoli: Canonical surfaces in P^5 and Calabi-Yau threefolds in P^6. Thesis Padova, pages 1–86, 2000.

35. C. Voisin: Courbes tétragonales et cohomologie de Koszul. J. Reine Angew.

Math., 387:111–121, 1988.

36. C. Walter: Pfaffian subschemes. J. Alg. Geom., 5:671–704, 1996.

Index

Brill-Noether number 3

Calabi-Yau 3-folds 17 canonical curve 13 Clifford index 13 Coble self-dual sets of points 13

differential operators 14 graded Artinian

Gorenstein ring 14 Green’s conjecture 13

Hartshorne-Rao module 6

Hilbert scheme 1

infinitesimal deformations 21

lift to characteristic zero 27 maximal rank 6 mirror families 27 moduli space 1 moduli space of curves 3 number field 27

Pfaffian complex 18 random curves 3 random nodal curve 4 random points 3 rank 1 linear syzygy 22

solving polynomial

equations 1 Syzygies of canonical curves 13

table of Betti numbers 2 unirationality 1, 8